

ODS TAGSETS.RTF: Tips and Tricks

Xiangxiang Meng, University of Cincinnati, Cincinnati, OH

ABSTRACT

In SAS 8.1 and later, the ODS RTF destination delivers and saves tables, listings, and figures in Rich Text Format (RTF) that can be read and edited by Microsoft Word 2002 and other word processing packages. New with SAS 9.2, ODS TAGSETS.RTF is created as a measured makeup to fix some problems with the traditional ODS RTF destination, such as memory consumption and page break restriction in the output of extremely large tables. This paper first introduces the common functions of the ODS RTF and ODS TAGSETS.RTF statements, and then illustrates some new features in the new ODS TAGSETS.RTF statement, such as controlling page break, adjusting table layout, and adding table of content in the RTF file.

INTRODUCTION

In SAS 8.1 and later, the ODS RTF destination delivers and saves tables, listings, and figures in Rich Text Format (RTF) that can be read and edited by Microsoft Word 2002 and other word processing packages. New with SAS 9.2, ODS TAGSETS.RTF is created as a measured makeup to fix some problems with the traditional ODS RTF destination, such as memory consumption and page break restriction in the output of extremely large tables. This paper first introduces the common functions of the ODS RTF and ODS TAGSETS.RTF statements, and then illustrates some new features in the new ODS TAGSETS.RTF statement, such as controlling page break, adjusting table layout, and adding table of content in the RTF file.

ODS TAGSETS.RTF V.S. ODS RTF

Here are some of the major differences between the ODS TAGSETS.RTF and the traditional ODS RTF destinations.

<u>ODS RTF</u>	<u>ODS TAGSETS.RTF</u>
<ul style="list-style-type: none">• Load the entire table into memory before being rendered• Let WORD determine page breaks when tables are longer than a physical page• RTF control string can only be used in TEXT= option• Based on the RTF specifications in Microsoft WORD	<ul style="list-style-type: none">• Only load about a page of data into the memory at one time.• Page breaks within long tables can be determined in SAS by using panels• Can use RTF control string for adjusting table layouts• Better support RTF readers other than Microsoft WORD

GETTING TO KNOW ODS TAGSET.RTF

The ODS TAGSETS.RTF statement has the same syntax as the traditional ODS RTF statement, and thus supports all the actions and most of the options that are available in ODS RTF. This section briefly summarizes some useful actions and options shared by both RTF statements.

SYNTAX

ODS TAGSETS.RTF (<ID=>Identifier) (<File=>file-name) action;

ODS TAGSETS.RTF (<ID=>Identifier) (<File=>file-name) <option(s)>;

OPENING AND CLOSING ODS RTF DESTINATIONS

In ODS TAGSETS.RTF, using FILE= option closes the current ODS RTF destination and the files associated with it, and then opens a new instance of the RTF destination with a specified file name. Without using the FILE= option, the ODS TAGSETS.RTF statement maintains and modifies the current ODS RTF destination.

```

* Example 1;
ods tagsets.rtf file='C:\example1_1.rtf' style=BarrettsBlue startpage=no;
proc print data=sashelp.cars label ;
  id make;
  var engineSize mpg_city mpg_highway weight length;
  where Make='Acura';
run;

ods tagsets.rtf style=watercolor;
proc print data=sashelp.cars label ;
  id make;
  var engineSize mpg_city mpg_highway weight length;
  where Make='Jeep';
run;

* no need to close the previous RTF;
ods tagsets.rtf file='C:\example1_2.rtf' style=sasweb;
proc print data=sashelp.cars label ;
  id make;
  var engineSize mpg_city mpg_highway weight length;
  where Make='Acura';
run;
ods tagsets.rtf close;

```

The first ODS TAGSETS.RTF statement automatically opens a new file 'C:\example1_1.rtf' and the second one change the ODS style of the current RTF destination. The third target statement closes the first file and then opens a new one. Unlike opening RTF files, an explicit CLOSE action is required to close the last RTF destination in the program.




Figure 1. The Two RTF Outputs in Example 1

OPENING MULTIPLE RTF FILES

Some applications require parallel output to multiple RTF files rather than the sequential output in the first example. These parallel outputs might contain different sets of tables generated from one SAS program, or have different styles and layouts. In both ODS RTF and ODS TAGSETS.RTF, multiple instances of the RTF destination can be opened at the same time by assigning IDs to the instances. Moreover, using the **SELECT** or **EXCLUDE** action determine whether a specific output will be delivered to a RTF file. The following example opens two RTF files at the same time and the outputs are the same as those in the first example:

```
* Example 2;
option nodate;
title; footnote;
ods tagsets.rtf(ID=file1) file='C:\example1_1_copy.rtf' startpage=NO;
ods tagsets.rtf(ID=file2) file='C:\example1_2_copy.rtf';
ods tagsets.rtf(ID=file1) style=BarrettsBlue;
ods tagsets.rtf(ID=file2) style=sasweb;

proc print data=sashelp.cars label ;
  id make;
  var engineSize mpg_city mpg_highway weight length;
  where Make='Acura';
run;

ods tagsets.rtf(ID=file1) style=watercolor;
ods tagsets.rtf(ID=file2) exclude all; /* or, include none */

proc print data=sashelp.cars label ;
  id make;
  var engineSize mpg_city mpg_highway weight length;
  where Make='Jeep';
run;

* NOTE: need to close the files separately;
ods tagsets.rtf(ID=file1) close;
ods tagsets.rtf(ID=file2) close;
```

Note:

- ODS TAGSETS.RTF EXCLUDE|SELECT has the same syntax as ODS SELECT|EXCLUDE.
- There are four actions in TAGSETS.RTF: CLOSE, SELECT, EXCLUDE, and SHOW. The SHOW action writes to the SAS log the current selection/exclusion list for the RTF destination.

OPTIONS IN ODS.TAGSETS.RTF

Most of the options in the ODS RTF statement are still supported by ODS TAGSETS.RTF. Some useful options are listed in the following table:

Option	Description
ANCHOR='anchor-name'	Specifies the base name for the RTF anchor tag that identifies each output object in current file
COLUMNS=	Specifies the number of columns to create in the output
FILE='external-file'/'fileref'	Opens the RTF destination and the associated RTF file.
GFOOTNOTE NOGFOOTNOTE	Includes or suppresses all titles in the current graphics output
GTITLE NOGTITLE	Includes or suppresses all titles in the current graphics output
NEWFILE=BYGROUP NONE OUTPUT PROC	Creates a new file at the specified starting-point such as by each group, procedure, and output
STYLE='style-definition'	Specifies the style in use. by default, use styles.RTF
STARTPAGE=YES NO NOW	Controls page breaks in the RTF file
TEXT='text-string'	Inserts text into the associated RTF file

The next two examples show how to use ODS TAGSETS.RTF options.

```
* Example 3;
ods tagsets.rtf file='C:\example3_1.rtf'
  nogtitle nogfootnote
  startpage=NO newfile=proc
;
proc univariate data=sashelp.cars;
  var mpg_city mpg_highway weight length;
run;
ods tagsets.rtf text='this is the end of the first file';

proc univariate data=sashelp.cars;
  var mpg_city mpg_highway weight length;
run;
ods tagsets.rtf text='This is the end of the second file';
ods tagsets.rtf close;
```

In example 3, a RTF file is created at the beginning of each procedure. The name of the first RTF file is specified, and the second RTF is automatically named 'example3_2.rtf'. Graphics titles and footnotes are suppressed, and there is no page break between tables. A text is inserted at the end of each file:




Figure 2. The Text Inserted at the End of Each RTF File by TEXT= Option

```

* Example 4 - Using anchors;
ods escapechar='^' noproctitle;
ods tagsets.rtf file='C:\example4_1.rtf'
  nogtitle nogfootnote newfile=none startpage=yes
;
ods tagsets.rtf text='This file includes listings for three car makers:';
ods tagsets.rtf text="^S={url='#proc1'} 1. Acura" ;
ods tagsets.rtf text="^S={url='#proc2'} 2. Jeep";
ods tagsets.rtf text="^S={url='#proc3'} 3. Scions" startpage=now;

ods tagsets.rtf anchor='proc1';
proc print data=sashelp.cars label noobs;
  var Model EngineSize mpg_city mpg_highway weight length;
  where Make='Acura';
run;

ods tagsets.rtf anchor='proc2';
proc print data=sashelp.cars label noobs;
  var Model EngineSize mpg_city mpg_highway weight length;
  where Make='Jeep';
run;

ods tagsets.rtf anchor='proc3';
proc print data=sashelp.cars label noobs;
  var Model EngineSize mpg_city mpg_highway weight length;
  where Make='Scion';
run;
ods tagsets.rtf close;

```

This example first declares “^” as the ODS escape char, and then creates a link to the output of each PROC PRINT using the escape sequence in the ODS TEXT string. ANCHOR= option is then used before each PROC PRINT as a place the associated URL would link to. In the RTF output, *CTRL Click* on the name of a car maker can jump to the associated listing.




Figure 3. Creating URLs with ANCHOR= Option

SEVERAL NEW FEATURES IN ODS TAGSETS.RTF

ADDING A TABLE OF CONTENT

When using the listing output, a table of content is created and displayed in the Result Tree Window. This is very helpful when there are a lot of outputs from multiple procedures. New with SAS 9.2, the ODS TAGSETS.RTF also supports this feature with sub-options CONTENTS= and TOC_DATA=. To distinguish from the options shared by both ODS TAGSETS.RTF and ODS RTF statements, a new “OPTIONS” option is introduced and contains some sub-options that are exclusive to TAGSETS.RTF. CONTENTS= and TOC_DATA= are two of these sub-options that insert a table of content at the beginning of a RTF file and determine which output to be included in the table of content.

```

* Example 5;
* Adding a Table of Content;
* CONTENTS= and TOC_DATA= are sub options;

ods tagsets.rtf file='example5_1.rtf' options(contents='yes' toc_data='yes');
proc univariate data=sashelp.cars;
    var msrp mpg_city weight ;
run;

proc corr data=sashelp.cars;
    var msrp mpg_city weight;
run;

ods tagsets.rtf options(toc_data='no');
proc glm data=sashelp.cars;
    model msrp = mpg_city weight;
run; quit;
ods tagsets.rtf close;

```

Note:

CONTENTS= is available in the ODS RTF as an option, and The TOC_DATA is new with ODS TAGSETS.RTF. Both options are placed into OPTIONS option because they cooperate most of the time.

- In the RTF output, click "Update Field" to update and display the table of content.

The Univariate Procedure	2
MSRP	2
Moments.....	2
Basic Measures of Location and Variability	2
Tests For Location.....	2
Quantiles.....	2
Extreme Observations.....	3
MPG_City	4
Moments.....	4
Basic Measures of Location and Variability	4
Tests For Location.....	4
Quantiles.....	4
Extreme Observations.....	4
Weight	6
Moments.....	6
Basic Measures of Location and Variability	6
Tests For Location.....	6
Quantiles.....	6
Extreme Observations.....	7
The Corr Procedure.....	8
Variables Information.....	8
Simple Statistics.....	8
Pearson Correlations.....	8

Figure 4. Creating a Table of Content in the RTF Output

PANELING IN LARGE TABLE OUTPUT

ODS TAGSETS.RTF and ODS RTF have different ways in organizing the output for a large table. For example:

```

* Example 6;
ods rtf file='C:\example6_1.rtf';
proc print data=sashelp.cars;
run;
ods rtf close;

ods tagsets.rtf file='C:\example6_2.rtf';
proc print data=sashelp.cars;
run;
ods tagsets.rtf close;

```

In the first file, ODS RTF outputs all the rows for the first seven variables and then all the rows for the next set of variables. In contrast, ODS TAGSETS.RTF print out all the variables for the first 17 rows in the first page, and then all the variables for the set of rows in the second page. The second way is more readable since it put all the information for one observation on the same page. The sets of observations automatically divided by ODS TAGSETS.RTF is usually referred to as panels.

Panels can be explicitly created and controlled by the `TABLEROWS=` and `PAGEPANELS=` options. The `TABLEROWS` options specify the number of rows (observations) for a panel, and the `PAGEPANELS` options determine the maximum number of panels per page. By default, the panels hold as many columns as can fit on one line. The panels continue until all the columns are produced for the current set of rows (specified by `TABLEROWS=`), and then a page break is forced before the output of the next set of rows.

```
* Example 7;
* 10 rows in a panel and at most 5 panels in a page;
ods tagsets.rtf file='C:\example7.rtf' tablerows=10 pagepanels=5;
proc print data=sashelp.cars;
run;
ods tagsets.rtf close;
```

Obs	Make	Model	Type	Origin	DriveTrain	MSRP	Invoice	EngineSize
1	Acura	MDX	SUV	Asia	All	\$36,945	\$33,337	3.5
2	Acura	RSX Type S 2dr	Sedan	Asia	Front	\$23,820	\$21,761	2.0
3	Acura	TSX 4dr	Sedan	Asia	Front	\$26,990	\$24,647	2.4
4	Acura	TL 4dr	Sedan	Asia	Front	\$33,195	\$30,299	3.2
5	Acura	3.5 RL 4dr	Sedan	Asia	Front	\$43,755	\$39,014	3.5
6	Acura	3.5 RL w/Navigation 4dr	Sedan	Asia	Front	\$46,100	\$41,100	3.5
7	Acura	NSX coupe 2dr manual S	Sports	Asia	Rear	\$89,765	\$79,978	3.2
8	Audi	A4 1.8T 4dr	Sedan	Europe	Front	\$25,940	\$23,508	1.8
9	Audi	A4 1.8T convertible 2dr	Sedan	Europe	Front	\$35,940	\$32,506	1.8
10	Audi	A4 3.0 4dr	Sedan	Europe	Front	\$31,840	\$28,846	3.0

Obs	Cylinders	Horsepower	MPG_City	MPG_Highway	Weight	Wheelbase	Length
1	6	265	17	23	4451	106	189
2	4	200	24	31	2778	101	172
3	4	200	22	29	3230	105	183
4	6	270	20	28	3575	108	186
5	6	225	18	24	3880	115	197
6	6	225	18	24	3893	115	197
7	6	290	17	24	3153	100	174
8	4	170	22	31	3252	104	179
9	4	170	23	30	3638	105	180
10	6	220	20	28	3462	104	179

(Continued)

Figure 5. Paneled Outputs with `TABLEROW=` and `PAGEPANEL` Options

USING RTF CONTROL STRINGS

Beginning with SAS release 8.2, RTF control words can be used within the `TITLE` and `FOOTNOTE` statements to insert formatting texts in titles and footnotes. RTF control words are defined by the Rich Text Format (RTF) Specification and serve as specially formatted commands for defining, managing and printing a RTF document. For example, the following statements produce titles with different text formats using control words `\i`, `\cfN`, and `\ulth`:

```
title1 '\i this is an italicized title';
title2 '\cf2 this is a tile with foreground color';
title3 '\ulth this title has thick underline';
```

The ODS TAGSETS.RTF statement has three new sub-options, TROWD=, TRHDR=, and TROWHDRCELL=, which allow direct specification of RTF control words for table formatting such as table alignment, text alignment, and header formatting. The following example shows how to use TROWD= options.

```

* Example 8;
* output the same tables with different alignments and row heights;

ods tagsets.rtf file='C:\example8.rtf' startpage=no
  options(TROWD='\trql \trrh500') ;
proc print data=sashelp.cars label noobs;
  var Model EngineSize mpg_city mpg_highway weight length;
  where Make='Jeep';
run;

ods tagsets.rtf options(TROWD='\trqc \trrh750');
proc print data=sashelp.cars label noobs;
  var Model EngineSize mpg_city mpg_highway weight length;
  where Make='Jeep';
run;

ods tagsets.rtf options(TROWD='\trqr \trrh1000');
proc print data=sashelp.cars label noobs;
  var Model EngineSize mpg_city mpg_highway weight length;
  where Make='Jeep';
run;

ods tagsets.rtf close;

```

The screenshot displays three SAS output tables, each representing the same data but with different formatting options applied via the TROWD= option. The data is as follows:

Model	Engine Size (L)	MPG (City)	MPG (Highway)	Weight (LBS)	Length (IN)
Grand Cherokee Laredo	4.0	16	21	3790	181
Liberty Sport	2.4	20	24	3826	174
Wrangler Sahara convertible 2dr	4.0	16	19	3575	150

The first table (top) is left-aligned and has a row height of 500. The second table (middle) is centered and has a row height of 750. The third table (bottom) is right-aligned and has a row height of 1000.

Figure 6. Adjusting table alignments and row heights with TROWD option

Next example shows how to adjust the header of the table with the TROWHDRCELL option:

```

* Example 9;
ods tagsets.rtf file='C:\example9_1.rtf' startpage=no
  options(TROWHDRCELL='\trrh2000 \clcbpat3 \clbgdcross \clvertalt' );
proc print data=sashelp.cars label noobs;
  var Model EngineSize mpg_city mpg_highway weight length;
  where Make='Jeep';
run;

ods tagsets.rtf
  options(TROWHDRCELL='\trrh2000 \clcbpat8 \clbgfdiag \clvertalb' );
proc print data=sashelp.cars label noobs;
  var Model EngineSize mpg_city mpg_highway weight length;
  where Make='Jeep';
run;

ods tagsets.rtf close;

```




Figure 7. Adjusting table header with TROWHDRCELL option

Some useful RTF control strings are summarized in the following table:

RTF Control String	Description
\trql \trqc \trlr	Left-justify, center, or right-justify a table or a table row
\rtlrow \ltrrow	Place the cells in the table row right to left or left to right
\trrhN	Control the height of a table row in N twips
\trleftN	Specify the position in twips of the leftmost edge of the table with respect to the left edge of its column.
\clcbpatN	N is the line color of the background pattern.

<code>\clcbpat</code> <i>N</i>	<i>N</i> is the background color of the background pattern.
<code>\clbg</code> <i>SSSS</i>	Specify a background pattern for the cell. For example SSSS=horiz horizontal lines. SSSS=vert vertical lines. SSSS=fdiag forward diagonal lines. (\\)\\ SSSS=bdiag backward diagonal lines. (///) SSSS=cross cross lines. (///)
<code>\clvertal</code> <i>S</i>	S= C, T and B for center/top/bottom alignment of the text in cells
<code>\cltxlrtbv</code>	Text in a cell flows left to right and top to bottom, vertical.
<code>\cltxtbrlv</code>	Text in a cell flows top to bottom and right to left, vertical.

REFERENCES

1. SAS 9.2 Output Delivery System: User's Guide
2. Hester, Wayne 2006. "Teaching Your RTF Tagset to Do Clever Tricks." Proceeding of the Thirty-first Annual SAS Users Group International Conference.

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the author at:

Xiangxiang Meng
Department of Mathematical Science University of Cincinnati
mengxa@mail.uc.edu

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration. Other brand and product names are registered trademarks or trademarks of their respective companies.